

Optimizing DiffEq Code

Chris Rackauckas

September 4, 2020

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.83 MiB
  allocs estimate: 101434
  -----
  minimum time:      3.317 ms (0.00% GC)
  median time:       3.379 ms (0.00% GC)
  mean time:         4.072 ms (16.30% GC)
  maximum time:     10.509 ms (44.14% GC)
  -----
  samples:           1227
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.47 MiB
  allocs estimate: 88652
  -----
  minimum time:      2.908 ms (0.00% GC)
  median time:       2.917 ms (0.00% GC)
  mean time:         3.438 ms (14.01% GC)
  maximum time:     8.360 ms (43.69% GC)
  -----
  samples:           1453
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to *****make our code non-allocating*****:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
  du[2] = u[1]*(28.0-u[3]) - u[2]
  du[3] = u[1]*u[2] - (8/3)*u[3]
end

lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes

the internal allocations as well. When we benchmark this function, we will see quite a difference.

```
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate:  1.38 MiB
  allocs estimate: 12996
  -----
  minimum time:     776.810 μs (0.00% GC)
  median time:      783.761 μs (0.00% GC)
  mean time:        871.667 μs (9.30% GC)
  maximum time:     5.647 ms (80.18% GC)
  -----
  samples:          5723
  evals/sample:     1

@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.22 KiB
  allocs estimate:  54
  -----
  minimum time:     341.247 μs (0.00% GC)
  median time:      344.138 μs (0.00% GC)
  mean time:        344.614 μs (0.00% GC)
  maximum time:     387.571 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1
```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```
tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  5.22 KiB
  allocs estimate:  54
  -----
  minimum time:     1.713 ms (0.00% GC)
  median time:      1.722 ms (0.00% GC)
  mean time:        1.725 ms (0.00% GC)
  maximum time:     2.008 ms (0.00% GC)
  -----
  samples:          2897
  evals/sample:     1
```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are](#)

a lot of sources online. But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. **structs** for example are stack-allocated "value-type"s. **Tuples** are a stack-allocated collection. The most useful data structure for DiffEq though is the **StaticArray** from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```
using StaticArrays
A = @SVector [2.0,3.0,5.0]

3-element StaticArrays.SArray{Tuple{3},Float64,1,3} with indices SOneTo(3):
 2.0
 3.0
 5.0
```

Notice that the 3 after **SVector** gives the size of the **SVector**. It cannot be changed. Additionally, **SVectors** are immutable, so we have to create a new **SVector** to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. **SArrays** have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons. Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's *****optimize lorenz using static arrays*****. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end

lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```

BenchmarkTools.Trial:
  memory estimate:  466.28 KiB
  allocs estimate:  2595
  -----
  minimum time:     337.296 μs (0.00% GC)
  median time:      341.729 μs (0.00% GC)
  mean time:        360.330 μs (4.14% GC)
  maximum time:     2.846 ms (84.09% GC)
  -----
  samples:          10000
  evals/sample:     1

@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  3.33 KiB
  allocs estimate:  28
  -----
  minimum time:     236.590 μs (0.00% GC)
  median time:      240.413 μs (0.00% GC)
  mean time:        240.963 μs (0.00% GC)
  maximum time:     277.322 μs (0.00% GC)
  -----
  samples:          10000
  evals/sample:     1

```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```

A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:     1.122 ms (0.00% GC)
  median time:      1.147 ms (0.00% GC)
  mean time:        1.290 ms (11.00% GC)

```

```

maximum time:      2.757 ms (56.80% GC)
-----
samples:           3859
evals/sample:      1

```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+ C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```

test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  8
-----
 minimum time:     2.786 ms (0.00% GC)
 median time:     2.815 ms (0.00% GC)
 mean time:       2.949 ms (4.50% GC)
 maximum time:    4.367 ms (34.96% GC)
-----
 samples:         1694
 evals/sample:    1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
 minimum time:     1.115 ms (0.00% GC)
 median time:     1.137 ms (0.00% GC)
 mean time:       1.281 ms (11.13% GC)
 maximum time:    2.748 ms (57.56% GC)
-----
 samples:         3884
 evals/sample:    1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2

```

```

-----
minimum time:      1.121 ms (0.00% GC)
median time:       1.146 ms (0.00% GC)
mean time:         1.287 ms (11.05% GC)
maximum time:      2.748 ms (57.78% GC)
-----
samples:           3868
evals/sample:      1

```

is a version with only 1 array created (the output). Note that `.s` can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```

1000×@*(1000 Array{@Float64,2}:
 0.969988 1.27378 0.674595 1.49687 ...@*( 0.575709 1.39178 0.2378321.00401 1.25521
1.50086 1.09863 0.445901 0.803779 1.162850.629866 0.69005 1.19046 0.849366 1.63814
1.00179 0.7254460.791673 1.11129 1.11416 0.680228 0.592147 1.42348 0.8549260.699215
0.968703 0.662798 0.65586 1.17703 1.28064 1.077931.11775 1.0985 1.18577 1.52222 (*@...@*(
0.829737 0.432762 0.1043991.09912 1.42333 1.10264 1.21103 0.365312 0.550715
0.7843230.935565 0.450125 0.524332 0.635121 0.993243 0.962044 0.9391140.634447 1.02329
0.553065 1.38375 0.547839 0.246321 0.5864810.543593 1.03386 1.15098 0.829146 1.41932
0.849664 0.721543(*@:@*( (*@`'.@*(1.32897 1.38425 0.612675 0.554495 0.688467 0.718432
1.056240.671591 1.12878 0.853041 1.3727 1.36853 1.43325 0.6536640.77173 1.01017 1.13647
0.303296 1.28936 1.00915 1.080071.48975 1.33525 0.967198 0.105555 0.878874 1.34308
0.6126391.34993 0.256636 0.828041 1.20592 (*@...@*( 1.5161 0.70398 0.9631350.873105
0.951314 1.0415 1.03118 1.48401 0.21581 1.213920.664695 0.826574 1.02836 1.01539 0.856151
0.573961 0.7111530.564979 1.061 0.489618 1.18759 0.534736 0.84394 0.77281.01794 1.60396
1.18406 1.5277 1.04868 0.966708 0.981435

```

Also, the `@.` macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
  allocs estimate:  2
-----
minimum time:      1.125 ms (0.00% GC)
median time:       1.145 ms (0.00% GC)
mean time:         1.290 ms (11.07% GC)
maximum time:      2.756 ms (57.51% GC)
-----
samples:           3859
evals/sample:      1

```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via `.=`. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

```

1000×@*(1000 Array{@Float64,2}:
 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 ...@*( 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 (*@...@*( 0.0

```

```

0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0(*@:*(
(*@:*( (*@`´.·.* ( (*@:*(0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 (*@...@*( 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0
0.0 0.00.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.00.0 0.0 0.0 0.0 0.0 0.0
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0

```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:

```

test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.044 ms (0.00% GC)
  median time:      1.051 ms (0.00% GC)
  mean time:        1.053 ms (0.00% GC)
  maximum time:     1.335 ms (0.00% GC)
  -----
  samples:          4725
  evals/sample:     1

```

If we use `@.` before the `=`, then it will turn it into `.=`:

```

test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:     1.046 ms (0.00% GC)
  median time:      1.054 ms (0.00% GC)
  mean time:        1.057 ms (0.00% GC)
  maximum time:     1.340 ms (0.00% GC)
  -----
  samples:          4707
  evals/sample:     1

```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```

test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)

```

```

BenchmarkTools.Trial:
  memory estimate:  32 bytes
  allocs estimate:  1
  -----
  minimum time:     2.296 ms (0.00% GC)
  median time:      2.305 ms (0.00% GC)

```



```

mean time:      2.308 ms (0.00% GC)
maximum time:   2.536 ms (0.00% GC)
-----
samples:        2163
evals/sample:   1

```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:      6.473 ms (0.00% GC)
median time:       6.609 ms (0.00% GC)
mean time:         6.736 ms (2.00% GC)
maximum time:      8.743 ms (18.52% GC)
-----
samples:           742
evals/sample:      1

```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```

using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B

```

```

BenchmarkTools.Trial:
 memory estimate:  0 bytes
 allocs estimate:  0
-----
minimum time:      6.123 ms (0.00% GC)
median time:       6.160 ms (0.00% GC)
mean time:         6.162 ms (0.00% GC)
maximum time:      6.722 ms (0.00% GC)
-----
samples:           811
evals/sample:      1

```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, `***`we can fuse together higher level linear algebra operations using BLAS`***`. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal{Float64,3}}{1.0 for i in 1:N-1, [-2.0 for i in 1:N], [1.0 for i in 1:N-1]}
Ay = copy{Ax}
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du .+ a.*u.*u./v .+ ubar .- α*u
    dr[:, :, 2] = Dv .+ a.*u.*u .- β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64}(100,100,2)
r0[:, :, 1] .= uss.+0.1.*rand{Float64}()
r0[:, :, 2] .= vss

prob = ODEProblem{basic_version!,r0,(0.0,0.1),p}

ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.057736107589742 11.041090915504622 .. @*( 11.081448552296811 11.093782388396566;
11.04648127284718 11.086110235651716 (*@...@*( 11.075788958028195 11.032707591876191;
(*@...@*( ; 11.032239058583906 11.033703155264943 (*@...@*(
11.07344801144967211.00740619405473; 11.010062475449605 11.008065921167299 (*@...@*(
11.07444694543734 11.02362082804462] [12.100000000000001 12.100000000000001 (*@...@*(
12.100000000000001 12.100000000000001; 12.100000000000001 12.100000000000001 (*@...@*(
12.100000000000001 12.100000000000001; (*@...@*( ; 12.100000000000001 12.100000000000001
(*@...@*( 12.100000000000001 12.100000000000001; 12.100000000000001 12.100000000000001
(*@...@*( 12.10000000000000112.100000000000001]
```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 194.52 MiB
  allocs estimate: 8899
  -----
  minimum time:      50.511 ms (3.90% GC)
  median time:       52.732 ms (7.39% GC)
  mean time:         52.569 ms (7.15% GC)
  maximum time:      52.988 ms (7.45% GC)
  -----
  samples:           96
```

```
evals/sample:      1
```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```
A = rand(4)
@show A
B = @view A[1:3]
B[2] = 2
@show A

A = [0.009042982926826149, 0.6747437463079622, 0.28229471973387166, 0.22312
941993947089]
A = [0.009042982926826149, 2.0, 0.28229471973387166, 0.22312941993947089]
4-element Array{Float64,1}:
 0.009042982926826149
 2.0
 0.28229471973387166
 0.22312941993947089
```

Notice that changing `B` changed `A`. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```
function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(A*y*u + u*Ax)
    Dv = D2*(A*y*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
 memory estimate: 124.44 MiB
  allocs estimate: 7369
  -----
 minimum time:      40.217 ms (4.84% GC)
 median time:      40.481 ms (4.93% GC)
 mean time:        41.079 ms (6.34% GC)
 maximum time:     42.689 ms (9.25% GC)
  -----
 samples:          122
 evals/sample:     1
```

Now, most of the allocations are taking place in $Du = D1*(Ay*u + u*Ax)$ since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```
Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 30.98 MiB
  allocs estimate: 5533
  -----
  minimum time:      34.804 ms (0.00% GC)
  median time:       34.946 ms (0.00% GC)
  mean time:         35.567 ms (1.74% GC)
  maximum time:      37.020 ms (5.16% GC)
  -----
  samples:            141
  evals/sample:       1
```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability*****.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
    a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
```

```

    @. du = Du + a*u*u./v + ubar -  $\alpha$ *u
    @. dv = Dv + a*u*u -  $\beta$ *v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

BenchmarkTools.Trial:
  memory estimate: 30.88 MiB
  allocs estimate: 1095
  -----
  minimum time:      28.011 ms (0.00% GC)
  median time:       28.890 ms (0.00% GC)
  mean time:         29.235 ms (2.13% GC)
  maximum time:      30.973 ms (6.11% GC)
  -----
  samples:           171
  evals/sample:      1

```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```

p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a, $\alpha$ ,ubar, $\beta$ ,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
    @inbounds for j in 2:N-1
        i = 1
        du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = N
        du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
    @inbounds for j in 2:N-1
        i = N
        du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
            a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
    end

    @inbounds for i in 2:N-1
        j = 1
        du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
            a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    end
end

```

```

end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
 memory estimate: 30.85 MiB
 allocs estimate: 484
-----
 minimum time:      8.437 ms (0.00% GC)
 median time:      8.569 ms (0.00% GC)
 mean time:        9.168 ms (6.59% GC)
 maximum time:     10.626 ms (17.51% GC)
-----
 samples:           546
 evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get

the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then **optimize our algorithm choice**. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```
prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate:  2.76 GiB
  allocs estimate:  41650
  -----
  minimum time:     1.328 s (40.24% GC)
  median time:      1.572 s (37.64% GC)
  mean time:        1.537 s (40.65% GC)
  maximum time:     1.678 s (30.54% GC)
  -----
  samples:          4
  evals/sample:     1
```

```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate:  58.35 MiB
  allocs estimate:  9695
  -----
  minimum time:     388.059 ms (0.00% GC)
  median time:      389.783 ms (0.52% GC)
  mean time:        389.827 ms (0.33% GC)
  maximum time:     391.351 ms (0.53% GC)
  -----
  samples:          13
  evals/sample:     1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate:  2.90 MiB
  allocs estimate:  74
  -----
  minimum time:     5.737 s (0.00% GC)
  median time:      5.737 s (0.00% GC)
  mean time:        5.737 s (0.00% GC)
```

```

maximum time:      5.737 s (0.00% GC)
-----
samples:           1
evals/sample:      1

```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```

BenchmarkTools.Trial:
  memory estimate:  278.99 MiB
  allocs estimate:  51134
  -----
  minimum time:     2.113 s (0.00% GC)
  median time:      2.141 s (1.39% GC)
  mean time:        2.163 s (2.41% GC)
  maximum time:     2.235 s (5.67% GC)
  -----
  samples:          3
  evals/sample:     1

```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  3.36 MiB
  allocs estimate:  42691
  -----
  minimum time:     2.075 s (0.00% GC)
  median time:      2.077 s (0.00% GC)
  mean time:        2.077 s (0.00% GC)
  maximum time:     2.077 s (0.00% GC)
  -----
  samples:          3
  evals/sample:     1

```

```
prob = ODEProblem(fast_gm!,r0,(0.0,500.0),p)
```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  4.06 MiB
  allocs estimate:  54715
  -----
  minimum time:     2.666 s (0.00% GC)
  median time:      2.671 s (0.00% GC)
  mean time:        2.671 s (0.00% GC)
  maximum time:     2.676 s (0.00% GC)
  -----
  samples:          2
  evals/sample:     1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is `CVODE_BDF` doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas `CVODE_BDF` is taking large steps. Additionally, the `GMRES` linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

This tutorial is part of the SciMLTutorials.jl repository, found at: <https://github.com/SciML/SciMLTutorials.jl>. For more information on doing scientific machine learning (SciML) with open source software, check out <https://sciml.ai/>.

To locally run this tutorial, do the following commands:

```
using SciMLTutorials
SciMLTutorials.weave_file("introduction", "03-optimizing_diffeq_code.jmd")
```

Computer Information:

```
Julia Version 1.4.2
Commit 44fa15b150* (2020-05-23 18:35 UTC)
Platform Info:
  OS: Linux (x86_64-pc-linux-gnu)
  CPU: Intel(R) Core(TM) i7-9700K CPU @ 3.60GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-8.0.1 (ORCJIT, skylake)
```

Environment:

```
JULIA_LOAD_PATH = /builds/JuliaGPU/DiffEqTutorials.jl:
JULIA_DEPOT_PATH = /builds/JuliaGPU/DiffEqTutorials.jl/.julia
JULIA_CUDA_MEMORY_LIMIT = 2147483648
JULIA_NUM_THREADS = 8
```

Package Information:

```
Status `~/builds/JuliaGPU/DiffEqTutorials.jl/tutorials/introduction/Project.toml`
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.5.0
[0c46a032-eb83-5123-abaf-570d42b7fbaf] DifferentialEquations 6.15.0
[65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 5.5.0
[91a5bcd-55d7-5caf-9e0b-520d859cae80] Plots 1.6.3
[90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.12.4
[c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 4.3.0
[37e2e46d-f89d-539d-b4ee-838fcccc9c8e] LinearAlgebra
```